BUG REPORT

Talat Odman and Yongtao Hu
Georgia Tech

August 25, 2009

Synopsis: In CMAQ, numerical instabilities may be created during advection calculations. This problem may occur when the maximum wind speed decelerates by more than 1/3 during the course of an output time step.

We were able to track this omission back to Version 4.2 of CAMQ so it is very likely that the omission existed since the very beginning. Therefore, many applications in the last 10 years must have suffered from the symptoms of this omission. It is amazing that a bug this serious went unnoticed for so long.

Description: The advection schemes used in CMAQ are explicit. As such, they are subject to the Courant-Friedrichs-Lewy (CFL) condition for stability. This condition requires that the time step be smaller than the ratio of grid spacing to wind speed. In CMAQ, the CFL condition is applied only in the horizontal plane. Since horizontal grid spacing is uniform, the maximum wind speed in the domain determines the advection time step. In general, the uppermost layers are excluded from the maximum wind speed determination. Advection is sub-cycled in those layers (i.e., applied consecutively several times before other processes are applied) to satisfy the CFL condition.

The wind fields are supplied to CMAQ by a meteorological input file at a certain frequency (usually once every hour, but applications with more frequent inputs are becoming more common). For the time steps between two consecutive CMAQ output records, the wind fields are interpolated from the records in the meteorological input file for that particular time period. As an example, suppose the output time step is 1 hour and the frequency of meteorological inputs is 1/30 minutes. The advection calculations during the first half of the hour involve the winds at the top of the hour and at half past the hour. During the second half of the hour, the winds at half past the hour and at the end of the hour are used for advection. Therefore, all three wind records (i.e., the top, the middle, and the bottom of the hour) must be checked for maximum wind speed in the determination of the appropriate time step. However, only the last record, i.e. the record at the bottom of the hour (or the top of the next hour), is considered in CMAQ, When the maximum wind speeds at the top of the hour and any meteorological input steps in between are larger than the maximum wind speed at the bottom of the hour, there is a chance to violate the CFL condition and introduce numerical instability.

The reason this bug does not lead to problems more often is a factor of safety applied to the time step. In CMAQ, the time step is made smaller than 0.75 times grid spacing over wind speed and even smaller to be an integer number of seconds that divides the output time step evenly. If the maximum wind speed does not decrease by more than 1/3 during the course of an output time step, the factor of safety circumvents any instability.

Remedy: This bug did not exist in MAQSIP, the *legendary* prototype for CMAQ. Here is how MAQSIP determined the maximum wind speed in subroutine GETSTEP[1].

```
      IF ( FIRSTIME ) THEN

         (Other calculations)
C.............   Read the starting wind field:

         CALL CURRSTEP( JDATE, JTIME, SDATE, STIME, WSTEP,
     &        FDATE, FTIME )

         CALL HCONTVEL SUBST_GRID_ID
     $        ( UWIND, VWIND, FDATE, FTIME, 0 ) ! do not interpolate
      END IF                    !  if firstime

C.......   Establish ending time for this Courant number calculation:

      EDATE = JDATE
      ETIME = JTIME
      CALL NEXTIME ( EDATE, ETIME, TSTEP )

      WMAX = -1.0

 11   CONTINUE          ! loop evaluating max wind speed for this TSTEP

      DO L = 1, NLAYS
         DO R = 1, NROWS
            DO C = 1, NCOLS+1

               WMAX = MAX( WMAX, ABS( UWIND( C,R,L ) )/DX1 )

            ENDDO
         ENDDO

         DO R = 1, NROWS+1
            DO C = 1, NCOLS

               WMAX = MAX( WMAX, ABS( VWIND( C,R,L ) )/DX2 )

            ENDDO
         ENDDO
      ENDDO

      IF ( SECSDIFF( FDATE, FTIME, EDATE, ETIME ) .GT. 0 ) THEN

         CALL NEXTIME( FDATE, FTIME, WSTEP )

         CALL HCONTVEL SUBST_GRID_ID
     $        ( UWIND, VWIND, FDATE, FTIME, 0 ) ! do not interpolate

         GO TO  11              ! to head of loop

      END IF                    ! if you need to read the next input
time step.
```

[1] This portion of the MAQSIP code was developed by Dr. Carlie J. Coats at MCNC.

Note that the arguments of subroutine HCONTVEL are slightly different than those in CMAQ.

The GO TO – CONTINUE loop labeled 11 assures consideration of all the records between (JDATE, JTIME) and (EDATE, ETIME) of the meteorological input file with time step WSTEP. Also note that there is no such loop in subroutine ADVSTEP of CMAQ.